Campos vetoriais em 3 dimensões; Integrais de Linha e de Superficie

Campos vetoriais em três dimensões são especificados em Maple como em duas dimensões,exceto naturalmente que existem 3 componentes que são funções de 3 variaveis:

```
> restart:with(linalg):
```

```
> vf:=[2*x,2*y,1];
```

Warning, new definition for norm

Warning, new definition for trace

$$vf := [2\,x,\, 2\,y,\, 1]$$

Para plotar um campo 3-dimensional, precisamos da biblioteca "plots" --

```
> with(plots):
```

```
> fieldplot3d(vf,x=-2..2,y=-2..2,z=-2..2);
```



(Aprenda a usar as opções: "constrained" sob projeção, "boxed" sob o eixo "Z" e cores).

Uma integral de linha em tres dimensões envolve integrar uma função contra **ds** ou tambem **F** dot **dr** ... Vamos ilustrar o último para o campo de vetores dado acima e para a hélice cônica:

```
> ch:=[t*cos(8*t),t*sin(8*t),t,t=0..2];
```

$$ch := [t\cos(8\,t),\, t\sin(8\,t),\, t,\, t = 0\,..\,2]$$

```
> spacecurve(ch);
```

Plotamos o campo de vetores e a curva juntos para obter uma ideia do que esperamos da integral de linha:

```
> F:=fieldplot3d(vf,x=-2..2,y=-2..2,z=-2..2):
```

```
> G:=spacecurve(ch):
```

```
> display3d({F,G});
```



Integrais de linha em 3 dimensões são análogas a aquelas de duas dimensões. Observe a similaridade entre o o seguinte procedimento e o procedimento anterior para integral de linha em duas dimensões:

> **lineint3d:=proc(vf,cv)**

> **Int(dotprod(subs(x=cv[1],y=cv[2],z=cv[3],vf), diff([cv[1],cv[2],cv[3]],t)),cv[4]);**

> **end;**

lineint3d :=
proc(vf, cv) Int(dotprod(subs($x = cv[1]$, $y = cv[2]$, $z = cv[3]$, vf), diff([$cv[1]$, $cv[2]$, $cv[3]$], t)), $cv[4$

> **with(linalg):**

> **lineint3d(vf,ch);**

[Maple Math]

> **value(");**

$$6$$

Positivo, como esperado.

INTEGRAIS DE SURPERFICIE :

Como na integral de linha, o passo fundamental no cálculo de uma integral de superficie é a parametrização da superficie -- como sempre usaremos **u** e **v** como parametros. Se a superficie ja é expressada como **z=f(x,y)** , a parametrização mais simples é: **x=u, y=v, z=f(u,v)** .

Por exemplo, encontrar o fluxo do campo de votores

```
> vf:=[2*x,2*y,1];
```

$$vf := [2\,x, 2\,y, 1]$$

atraves da parte do paraboloide **z=1-x^2-y^2** acima do disco unitario **x^2+y^2<=1** .

Usamos a parametrização simples:

```
> surf:=[u,v,1-u^2-v^2];
```

$$surf := [u, v, 1 - u^2 - v^2]$$

com a variação :

```
> rg:=[v=-sqrt(1-u^2)..sqrt(1-u^2),u=-1..1];
```

$$rg := [v = -\sqrt{1-u^2} \,..\, \sqrt{1-u^2}, u = -1 \,..\, 1]$$

O grafico:

```
> F:=fieldplot3d(vf,x=-1..1,y=-1..1,z=-0..1):
```

```
> G:=plot3d(surf,u=-1..1,v=-1..1):
```

```
> display({F,G});
```



O elemento de area de superficie (vetorial) é:

```
> dsigma:=crossprod(diff(surf,u),diff(surf,v));
```

$$dsigma := [2\,u,\,2\,v,\,1]$$

e assim a integral de **vf** dot **normal unitaria d(area)** é

```
> Int(Int(dotprod(subs(x=surf[1],y=surf[2],z=surf[3],vf),dsigma),rg[1]),rg[2]);
```

$$\int_{-1}^{1}\int_{-\sqrt{1-u^2}}^{\sqrt{1-u^2}} 1+4\,u^2+4\,v^2\,dv\,du$$

```
> int(int(dotprod(subs(x=surf[1],y=surf[2],z=surf[3],vf),dsigma),rg[1]),rg[2]);
```

$$3\,\pi$$

>

Mais sobre integral de superficies nos proximos exemplos...